

<b>Dossier</b> <b>5</b>	<b>Ristournes annuelles</b>	
	Fonction : SI – Fonction : Arrondi Mise en forme conditionnelle	

<b>Apprentissage programmé</b>	
<b>Société Jurarbois</b>	
<b>Savoir-faire</b>	<ul style="list-style-type: none"> <li>• Programmer des formules conditionnelles</li> <li>• Utiliser la fonction Arrondi</li> <li>• Programmer une mise en forme conditionnelle</li> </ul>
<b>Pré-requis</b>	• Avoir réalisé les exercices du dossier 2
<b>Supports</b>	• Dossier <b>Jurarbois</b> sur la clé USB ou dans l'espace élève du disque dur ou du serveur
<b>Durée</b>	• 50 minutes

### Réalisation :

Dans ce dossier nous allons saisir le tableau suivant et programmer le calcul des ristournes annuelles sur le chiffre d'affaires réalisés avec les restaurateurs

- > Charger Excel
- > Saisir et mettre en forme le tableau suivant dans un nouveau classeur :

	A	B	C	D	E	F	G	H	I	J
1	<b>Ristourne annuelle aux restaurateurs</b>									
2	Restaurants	CA 2007	CA Blanc 2008	CA rouge 2008	CA Divers 2008	Total 2008	Ristourne 1	Ecart 2007-2008	Ristourne 2	Total Ristourne
3	Jaquin paul	1 820 €	1 225 €	830 €	425 €					
4	Moirier Julie	2 210 €	810 €	710 €	315 €					
5	Tromban Emile	1 450 €	450 €	620 €	120 €					
6	Carrier Paul	2 300 €	1 320 €	1 020 €	210 €					
7	Long Claud	1 950 €	980 €	830 €	410 €					
8	Carnon Pierre	260 €	460 €	660 €	250 €					
9	Halphan Luce	840 €	610 €	530 €	330 €					
10	Remy Tania	3 590 €	1 420 €	1 000 €	620 €					
11	Lacs Julie	1 020 €	320 €	520 €	150 €					
12	Rousset Camille	560 €	490 €	310 €	270 €					
13	<b>Totaux</b>									

#### ■ Enregistrer le classeur

- > Cliquer l'outil 
- > Sélectionner l'unité puis le dossier **Jurarbois**
- > Saisir le nom : **Ristournes annuelles**
- > Cliquer : **Enregistrer**

#### ■ Programmer le total 2008

- > Cliquer la cellule **F3**
- > Saisir la formule =**Somme(C3:E3)**
- > Recopier la formule sur les cellules **F4 à F12**

	A	B	C	D	E	F	G	H	I	J
1	<b>Ristourne annuelle aux restaurateurs</b>									
2	Restaurants	CA 2007	CA Blanc 2008	CA rouge 2008	CA Divers 2008	Total 2008	Ristourne 1	Ecart 2007-2008	Ristourne 2	Total Ristourne
3	Jaquin paul	1 820 €	1 225 €	830 €	425 €	2 480 €		660 €		
4	Moirier Julie	2 210 €	810 €	710 €	315 €	1 835 €		- 375 €		
5	Tromban Emile	1 450 €	450 €	620 €	120 €	1 190 €		- 260 €		
6	Carrier Paul	2 300 €	1 320 €	1 020 €	210 €	2 550 €		250 €		
7	Long Claud	1 950 €	980 €	830 €	410 €	2 220 €		270 €		
8	Carnon Pierre	260 €	460 €	660 €	250 €	1 370 €		1 110 €		
9	Halphan Luce	840 €	610 €	530 €	330 €	1 470 €		630 €		
10	Remy Tania	3 590 €	1 420 €	1 000 €	620 €	3 040 €		- 550 €		
11	Lacs Julie	1 020 €	320 €	520 €	150 €	990 €		- 30 €		
12	Rousset Camille	560 €	490 €	310 €	270 €	1 070 €		510 €		
13	<b>Totaux</b>									


#### ■ Programmer l'écart 2007-2008

- > Cliquer la cellule **H3**
- > Saisir la formule =**F3-B3**
- > Recopier la formule sur les cellules **H4 à H12**

Nous allons programmer l'affichage de telle sorte que les données positives soient mises sur un fond rouge.

### 1. Programmer une mise en forme conditionnelle

- > Sélectionner les cellules **H3 à J12** par cliqué glissé
- > Cliquer l'onglet **Accueil**
- > Cliquer l'outil  - Règle de mise en surbrillance des cellules
- > Cliquer **Supérieur à**
- > Saisir **0** et sélectionner la mise en forme désirée
- > Cliquer **OK**



Pour paramétrer une mise en forme conditionnelle différente sélectionner l'option : **Format personnalisé...**

## 2. Programmer une fonction conditionnelle



Nous allons programmer le calcul d'une ristourne de 5 % du chiffre d'affaires 2008 si ce dernier est supérieur à 1 500 € sinon la ristourne est de 2 %.

Nous allons utiliser une fonction conditionnelle dont la syntaxe est la suivante :

**=SI(condition;faire;sinon)**

=>

**=SI(F3>1500;F3\*5%;F3\*2%)**

travail à faire  
si condition remplie

travail à faire  
si condition non remplie

travail à faire  
si condition remplie

travail à faire  
si condition non remplie

Excel propose un assistant qui facilite la saisie des fonctions. Nous allons tester dans le cadre de ce dossier les deux méthodes (avec et sans assistant).

### 2.1 Programmer une fonction SI simple

-> Cliquer l'onglet : **Formules**

-> Pointer la cellule **G3** qui doit recevoir la ristourne 1

-> Cliquer le bouton puis l'option **SI**

=> La fenêtre de paramétrage apparaît :

-> Cliquer dans la zone : **Test\_logique**

-> Saisir la condition : **F3>1500**

-> Cliquer dans la zone : **Valeur\_si\_vrai**

-> Cliquer la cellule **F3** et saisir **\*5%**

-> Cliquer dans la zone : **Valeur\_si\_faux**

-> Cliquer la cellule **F3** et saisir **\*2%**

-> **OK** ou **[Entrée]**

-> Recopier la formule sur les cellules **H4 à H12**

=> Les résultats apparaissent dans les cellules et la formule dans la barre des formules.



### Programmer la fonction sans assistant

-> Pointer la cellule **G3**

-> Taper **=SI(**

-> Saisir au clavier **F3>1500**

-> Taper le signe ;

-> Saisir au clavier **F3\*5%**

-> Taper le signe ;

-> Saisir au clavier **F3\*2%**

-> Fermer la parenthèse )

-> **[Entrée]**

Ristourne annuelle aux restaurateurs									
	CA Blanc	CA rouge	CA Divers	Total	Ristourne	Ecart	Ristourne	Total	
2007	2008	2008	2008	2008	1	2007-2008	2	Ristourne	
3	Jaquin paul	1 820 €	1 225 €	830 €	425 €	2 480 €	124,00 €	660 €	
4	Moirier Julie	2 210 €	810 €	710 €	315 €	1 835 €	91,75 €	- 375 €	
5	Tromban Emile	1 450 €	450 €	620 €	120 €	1 190 €	23,80 €	- 260 €	
6	Carrier Paul	2 300 €	1 320 €	1 020 €	210 €	2 550 €	127,50 €	250 €	
7	Long Claud	1 950 €	980 €	830 €	410 €	2 220 €	111,00 €	270 €	
8	Carnon Pierre	260 €	460 €	660 €	250 €	1 370 €	27,40 €	1 110 €	
9	Halphan Luce	840 €	610 €	530 €	330 €	1 470 €	29,40 €	630 €	
10	Remy Tania	3 590 €	1 420 €	1 000 €	620 €	3 040 €	152,00 €	- 550 €	
11	Lacs Julie	1 020 €	320 €	520 €	150 €	990 €	19,80 €	- 30 €	
12	Rousset Camille	560 €	490 €	310 €	270 €	1 070 €	21,40 €	510 €	
13	<b>Totaux</b>								



### La condition peut être :

- un test alphabétique

**=Si(A2="Cadre";500;700)**

- un test numérique

**=Si(A2>1000;500;700)**

- un test sur formule de calcul

**=Si(Somme(A2:A10)>1000;500;700)**

### Le travail à faire peut être :

- l'affichage d'un nombre :

**=Si(A2>1000;500;700)**

- l'affichage d'un texte

**=Si(A2>1000;"Oui";"Non")**

- une formule de calcul

**=Si(A2>1000;A2\*5/100;A2\*10/100)**

- une fonction de calcul

**=Si(A2>1000;SOMME(C3:C5);SOMME(D3:D5))**

### SI Imbriqué

La fonction SI permet deux résultats. Pour en obtenir plus il faut imbriquer des conditions SI.

Exemple : une société accorde une réduction de 5 % si le CA est < 1000 ; 10 % si le CA est compris entre 1 000 et 2 000 et 15 % au dessus de 2 000. Solution : **=SI(CA<1000;5%;SI(CA>2000;15%;10%))**

### SI ET

La fonction **SI ET** permet de programmer la réalisation d'un travail sur la validité de plusieurs conditions.

Exemple : une remise de 10 % est accordée si le CA est > à 2000 **ET** si le nombre de commandes est supérieur à 10. Solution : **SI(ET(CA>2000;commande>10);10%;0%)**

### SI OU

La fonction **SI OU** permet de programmer la réalisation d'un travail sur la validité d'une condition sur plusieurs.

Exemple : une remise de 10 % est accordée si le CA est > à 2 000 **OU** si le nombre de commandes est supérieur à 10. Solution : **SI(OU(CA>2000;commande>10);10%;0%)**

Nous allons programmer la formule de la ristourne 2 qui fait appelle à deux conditions reliées par l'opérateur ET

**Calculer une ristourne de 2 % SI le Total 2008 est > à 2000 ET si l'Ecart 2007-2008 est > 0**

Cette fonction particulière est peu pratique à programmer avec l'assistant. Nous allons donc la programmer directement au clavier.

## 2.2 Programmer une fonction SI ET

- > Pointer la cellule I3 qui doit recevoir la ristourne 2
  - > Saisir : **SI(ET(**
  - > Saisir la condition 1 : **H3>0**
  - > Saisir ;
  - > Saisir la condition 2 : **F3>2000**
  - > Fermer la parenthèse 2 : **)**
  - > Saisir ;
  - > Saisir la travail à Faire si les conditions sont remplies : **F3\*2%**
  - > Saisir ;
  - > Saisir la travail à Faire si les conditions ne sont pas remplies : **0**
  - > Fermer la parenthèse 1 : **)**
  - > **OK** ou **[Entrée]**
  - > Recopier la formule sur les cellules I4 à I12
- => Les résultats apparaissent dans les cellules et la formule dans la barre des formules.

Ristourne annuelle aux restaurateurs										
Restaurants	CA 2007	CA Blanc 2008	CA rouge 2008	CA Divers 2008	Total 2008	Ristourne 1	Ecart 2007-2008	Ristourne 2	Total Ristourne	
Jaquin paul	1 820 €	1 225 €	830 €	425 €	2 480 €	124,00 €	660 €	49,60 €		
Moirier Julie	2 210 €	810 €	710 €	315 €	1 835 €	91,75 €	- 375 €	- €		
Tromban Emile	1 450 €	450 €	620 €	120 €	1 190 €	23,80 €	- 260 €	- €		
Carrier Paul	2 300 €	1 320 €	1 020 €	210 €	2 550 €	127,50 €	250 €	51,00 €		
Long Claud	1 950 €	980 €	830 €	410 €	2 220 €	111,00 €	270 €	44,40 €		
Carnon Pierre	260 €	460 €	660 €	250 €	1 370 €	27,40 €	1 110 €	- €		
Halphan Luce	840 €	610 €	530 €	330 €	1 470 €	29,40 €	630 €	- €		
Remy Tania	3 590 €	1 420 €	1 000 €	620 €	3 040 €	152,00 €	- 550 €	- €		
Lacs Julie	1 020 €	320 €	520 €	150 €	990 €	19,80 €	- 30 €	- €		
Rousset Camille	560 €	490 €	310 €	270 €	1 070 €	21,40 €	510 €	- €		
<b>Totaux</b>										

## 3. Calculer la ristourne total et l'arrondir



Nous allons calculer et arrondir la ristourne totale qui est égale à la ristourne 1 plus la ristourne 2. La formule de calcul de la cellule J3 est donc la suivante **=G3+I3**. Nous souhaitons arrondir la ristourne à l'entier le plus près, en conséquence la formule est intégrée dans une fonction **arrondi()** et la formule devient : **=arrondi(G3+I3;0)** ; 0 = nombre de décimal après la virgule

La fonction **Ent()** abouti au même résultat.**=ent(G3+I3)**

- > Cliquer la cellule J3
- > Taper **=Arrondi(G3+I3;0)**
- > **[Entrée]**
- > Recopier la formule sur les cellules J4 à J12

Ristourne annuelle aux restaurateurs										
Restaurants	CA 2007	CA Blanc 2008	CA rouge 2008	CA Divers 2008	Total 2008	Ristourne 1	Ecart 2007-2008	Ristourne 2	Total Ristourne	
Jaquin paul	1 820 €	1 225 €	830 €	425 €	2 480 €	124,00 €	660 €	49,60 €	174,00 €	
Moirier Julie	2 210 €	810 €	710 €	315 €	1 835 €	91,75 €	- 375 €	- €	92,00 €	
Tromban Emile	1 450 €	450 €	620 €	120 €	1 190 €	23,80 €	- 260 €	- €	24,00 €	
Carrier Paul	2 300 €	1 320 €	1 020 €	210 €	2 550 €	127,50 €	250 €	51,00 €	179,00 €	
Long Claud	1 950 €	980 €	830 €	410 €	2 220 €	111,00 €	270 €	44,40 €	155,00 €	
Carnon Pierre	260 €	460 €	660 €	250 €	1 370 €	27,40 €	1 110 €	- €	27,00 €	
Halphan Luce	840 €	610 €	530 €	330 €	1 470 €	29,40 €	630 €	- €	29,00 €	
Remy Tania	3 590 €	1 420 €	1 000 €	620 €	3 040 €	152,00 €	- 550 €	- €	152,00 €	
Lacs Julie	1 020 €	320 €	520 €	150 €	990 €	19,80 €	- 30 €	- €	20,00 €	
Rousset Camille	560 €	490 €	310 €	270 €	1 070 €	21,40 €	510 €	- €	21,00 €	
<b>Totaux</b>										

Il nous reste à programmer les totaux de colonnes

- > Cliquer la cellule **B13**
- > Cliquer l'outil Somme 
- > Sélectionner les cellules **B3 à B12**
- > **[Entrée]**
- > Recopier la formule sur les cellules de droite

	A	B	C	D	E	F	G	H	I	J
1	<b>Ristourne annuelle aux restaurateurs</b>									
2	Restaurants	CA 2007	CA Blanc 2008	CA rouge 2008	CA Divers 2008	Total 2008	Ristourne 1	Ecart 2007-2008	Ristourne 2	Total Ristourne
3	Jaquin paul	1 820 €	1 225 €	830 €	425 €	2 480 €	124,00 €	660 €	49,60 €	174,00 €
4	Moirier Julie	2 210 €	810 €	710 €	315 €	1 835 €	91,75 €	- 375 €	- €	92,00 €
5	Tromban Emile	1 450 €	450 €	620 €	120 €	1 190 €	23,80 €	- 260 €	- €	24,00 €
6	Carrier Paul	2 300 €	1 320 €	1 020 €	210 €	2 550 €	127,50 €	250 €	51,00 €	179,00 €
7	Long Claud	1 950 €	980 €	830 €	410 €	2 220 €	111,00 €	270 €	44,40 €	155,00 €
8	Carnon Pierre	260 €	460 €	660 €	250 €	1 370 €	27,40 €	1 110 €	- €	27,00 €
9	Halphan Luce	840 €	610 €	530 €	330 €	1 470 €	29,40 €	630 €	- €	29,00 €
10	Remy Tania	3 590 €	1 420 €	1 000 €	620 €	3 040 €	152,00 €	- 550 €	- €	152,00 €
11	Lacs Julie	1 020 €	320 €	520 €	150 €	990 €	19,80 €	- 30 €	- €	20,00 €
12	Rousset Camille	560 €	490 €	310 €	270 €	1 070 €	21,40 €	510 €	- €	21,00 €
13	<b>Totaux</b>	<b>16 000 €</b>	<b>8 085 €</b>	<b>7 030 €</b>	<b>3 100 €</b>	<b>18 215 €</b>	<b>728 €</b>	<b>2 215 €</b>	<b>145 €</b>	<b>873 €</b>

Nous avons terminé nous allons imprimer les formules de calcul.

### . Afficher et imprimer les formules de calcul

- > Cliquer l'onglet **Formules**
- > Cliquer l'outil  **Afficher les formules**
- > Réduire éventuellement la largeur des colonnes
- > Cliquer l'onglet **Mise en page**
- > Cliquer l'outil **Orientation**  et sélectionner **Paysage**
- > Cliquer l'outil 

	A	B	C	D	E	F	G	H	I	J
1	<b>Ristourne annuelle aux restaurateurs</b>									
2	Restaurants	CA 2007	CA Blanc 2008	CA rouge 2008	CA Divers 2008	Total 2008	Ristourne 1	Ecart 2007-2008	Ristourne 2	Total Ristourne
3	Jaquin paul	1820	1225	830	425	=SOMME(C3:E3)	=SI(F3>1500;F3*5%;F3*2%)	=F3-B3	=SI(ET(H3>0;F3>2000);F3*2%;0)	=ARRONDI(G3+H3;0)
4	Moirier Julie	2210	810	710	315	=SOMME(C4:E4)	=SI(F4>1500;F4*5%;F4*2%)	=F4-B4	=SI(ET(H4>0;F4>2000);F4*2%;0)	=ARRONDI(G4+H4;0)
5	Tromban Emile	1450	450	620	120	=SOMME(C5:E5)	=SI(F5>1500;F5*5%;F5*2%)	=F5-B5	=SI(ET(H5>0;F5>2000);F5*2%;0)	=ARRONDI(G5+H5;0)
6	Carrier Paul	2300	1320	1020	210	=SOMME(C6:E6)	=SI(F6>1500;F6*5%;F6*2%)	=F6-B6	=SI(ET(H6>0;F6>2000);F6*2%;0)	=ARRONDI(G6+H6;0)
7	Long Claud	1950	980	830	410	=SOMME(C7:E7)	=SI(F7>1500;F7*5%;F7*2%)	=F7-B7	=SI(ET(H7>0;F7>2000);F7*2%;0)	=ARRONDI(G7+H7;0)
8	Carnon Pierre	260	460	660	250	=SOMME(C8:E8)	=SI(F8>1500;F8*5%;F8*2%)	=F8-B8	=SI(ET(H8>0;F8>2000);F8*2%;0)	=ARRONDI(G8+H8;0)
9	Halphan Luce	840	610	530	330	=SOMME(C9:E9)	=SI(F9>1500;F9*5%;F9*2%)	=F9-B9	=SI(ET(H9>0;F9>2000);F9*2%;0)	=ARRONDI(G9+H9;0)
10	Remy Tania	3590	1420	1000	620	=SOMME(C10:E10)	=SI(F10>1500;F10*5%;F10*2%)	=F10-B10	=SI(ET(H10>0;F10>2000);F10*2%;0)	=ARRONDI(G10+H10;0)
11	Lacs Julie	1020	320	520	150	=SOMME(C11:E11)	=SI(F11>1500;F11*5%;F11*2%)	=F11-B11	=SI(ET(H11>0;F11>2000);F11*2%;0)	=ARRONDI(G11+H11;0)
12	Rousset Camille	560	490	310	270	=SOMME(C12:E12)	=SI(F12>1500;F12*5%;F12*2%)	=F12-B12	=SI(ET(H12>0;F12>2000);F12*2%;0)	=ARRONDI(G12+H12;0)
13	<b>Totaux</b>	=SOMME(B3:B12)	=SOMME(C3:C12)	=SOMME(D3:D12)	=SOMME(E3:E12)	=SOMME(F3:F12)	=SOMME(G3:G12)	=SOMME(H3:H12)	=SOMME(I3:I12)	=SOMME(J3:J12)

- > Cliquer l'outil  **Afficher les formules** pour afficher de nouveau les données
- > Cliquer l'outil **Orientation**  et sélectionner **Portrait**
- > Modifier la largeur des colonnes pour afficher toutes les données

Nous pouvons imprimer et sauvegarder la feuille de calcul.

### ▪ Imprimer la feuille

- > Cliquer l'outil  de la barre d'outils Accès rapide (pour une impression directe sans paramétrage)

### ▪ Sauvegarder le classeur

- > Cliquer l'outil 

### ▪ Quitter le classeur

- > Cliquer la case  fermeture du Classeur

<b>Dossier</b> <b>5</b>	<b>Indemnités de retard</b>	
	Fonction : SI – Fonction : Arrondi Mise en forme conditionnelle	

<b>EXERCICE EN AUTONOMIE</b>	
<b>Société Strap SA</b>	
<b>Savoir-faire</b>	<ul style="list-style-type: none"> <li>Saisir et mettre en forme un tableau avec formules de calcul et fonction conditionnelle</li> <li>Mise en forme conditionnelle</li> </ul>
<b>Pré-requis</b>	<ul style="list-style-type: none"> <li>Avoir effectué la visite guidée</li> </ul>
<b>Supports</b>	<ul style="list-style-type: none"> <li>Dossier <b>Strap SA</b> sur la clé USB ou dans l'espace élève du disque dur ou du serveur</li> </ul>
<b>Durée</b>	<ul style="list-style-type: none"> <li>40 minutes</li> </ul>

### Travail à Faire

1 - Charger Excel puis saisir et mettre en forme le tableau suivant

	A	B	C	D	E	F	G	H	I
	N° Appel d'offre	Chantier	Total Devis	Date début	Date fin	Fin réelle	Retard	Indemnités Retard	Remarques
2	286A	Citerne Alby	156 000,00 €	01/03/2008	30/04/2008	28/04/2008			
3	4568B	Four Montmin	15 000,00 €	01/04/2008	15/04/2008	22/04/2008			
4	1278A	Ecole de Giez	25 600,00 €	12/05/2008	15/06/2008	13/06/2008			
5	258CD	Citerne Annecy	32 000,00 €	15/06/2008	15/08/2008	15/08/2008			
6	478F	HLM Faverges	145 000,00 €	01/06/2008	01/07/2008	16/07/2008			
7	15915D	Parking Viuz	52 000,00 €	01/09/2008	01/10/2008	10/10/2008			
8	178A	Ecole Baland	78 000,00 €	15/07/2008	07/08/2008	05/08/2008			
9	56HRT	Rond point Hery	83 000,00 €	18/09/2008	18/10/2008	14/10/2008			
10	789 DEF	Ecole Faverges	45 000,00 €	20/09/2008	20/10/2008	28/10/2008			
11	129 MP	Citerne La Clusaz	140 000,00 €	20/09/2008	20/11/2008	15/11/2008			

2 - Programmer les calculs suivants :

- **Retard** = Fin réelle moins la date prévisionnelle de fin
- **Indemnité** = 0,1 % du montant total du chantier par jour de retard si le retard est positif
- **Remarques** = Si une indemnité est à verser inscrire le commentaire : Remettre le dossier à l'avocat

3 – Paramétrer une mise en évidence des retards positifs

4 - Sauvegarder le classeur dans le dossier **Strap SA** sous le nom **Indemnités de retard**

5 - Imprimer la feuille complète

6 - Imprimer les formules de calcul

7 - Quitter Excel

### Elément de correction : Tableau terminé

	A	B	C	D	E	F	G	H	I
	N° Appel d'offre	Chantier	Total Devis	Date début	Date fin	Fin réelle	Retard	Indemnités Retard	Remarques
2	286A	Citerne Alby	156 000,00 €	01/03/2008	30/04/2008	28/04/2008	-2	- €	
3	4568B	Four Montmin	15 000,00 €	01/04/2008	15/04/2008	22/04/2008	7	105,00 €	Remettre le dossier à l'avocat
4	1278A	Ecole de Giez	25 600,00 €	12/05/2008	15/06/2008	13/06/2008	-2	- €	
5	258CD	Citerne Annecy	32 000,00 €	15/06/2008	15/08/2008	15/08/2008	0	- €	
6	478F	HLM Faverges	145 000,00 €	01/06/2008	01/07/2008	16/07/2008	15	2 175,00 €	Remettre le dossier à l'avocat
7	15915D	Parking Viuz	52 000,00 €	01/09/2008	01/10/2008	10/10/2008	9	468,00 €	Remettre le dossier à l'avocat
8	178A	Ecole Baland	78 000,00 €	15/07/2008	07/08/2008	05/08/2008	-2	- €	
9	56HRT	Rond point Hery	83 000,00 €	18/09/2008	18/10/2008	14/10/2008	-4	- €	
10	789 DEF	Ecole Faverges	45 000,00 €	20/09/2008	20/10/2008	28/10/2008	8	360,00 €	Remettre le dossier à l'avocat
11	129 MP	Citerne La Clusaz	140 000,00 €	20/09/2008	20/11/2008	15/11/2008	-5	- €	

<b>Dossier</b>  <b>5</b>	<b>Primes sur salaires</b>	
	Fonction : SI – Fonction : Arrondi Mise en forme conditionnelle	

<b>Mise en situation</b> <b>Société Morisse Bois et dérivés</b>	
<b>Savoir-faire</b>	• Saisir et mettre en forme un tableau avec formules de calcul et fonction conditionnelle
<b>Pré-requis</b>	• Avoir effectué la visite guidée
<b>Supports</b>	• Dossier <b>Morisse bois</b> sur la clé USB ou dans l'espace élève du disque dur ou du serveur
<b>Durée</b>	• 45 minutes

### Travail à faire

Vous travaillez pour la société **Morisse bois et dérivés**. Cette société est spécialisée dans le traitement et la commercialisation de bois sous différentes formes :

- **Bois brut** ; planche, poutre, liteau etc.
- **Parquets** : tout type de bois
- **Accessoires** : vis, clous, colles etc.

Elle emploie sept représentants qui couvrent toute la France. Chaque mois le service commercial doit préparer les paye et remplir les tableaux concernant la rémunération et les indemnités kilométriques à prendre en compte pour réaliser les bulletins de salaire.

Le Directeur commercial, **Albert Gauchet**, vous remet

1. le tableau récapitulatif des ventes réalisées par les différents commerciaux au cours du mois de Mars. Ce tableau sert à calculer les primes et salaires bruts des représentants.
2. Le tableau des kilométrages et de la puissance fiscale des voitures utilisées par les représentants.

Représentants	CA Bois brut	CA Parquet	CA Accessoires
Pailler Luc	15356	12450	2350
Fardot Paul	12240	8560	4220
Campion Jean	8020	7700	2700
Gellen Lucie	7685	4200	2500
Collomb Irène	15230	9400	4500
Bataille Lucien	19700	12800	6500
Foutrot Emilie	14800	8200	7200
Représentants	Voiture CV	Kilométrage	
Pailler Luc	10	1200	
Fardot Paul	8	1800	
Campion Jean	12	1425	
Gellen Lucie	10	987	
Collomb Irène	10	810	
Bataille Lucien	12	1350	
Foutrot Emilie	8	1410	

### ➤ Tableau des salaires

Chaque mois les salariés perçoivent un salaire minimum de base de 1500 € auquel doit être ajouté une prime sur chiffre d'affaires et éventuellement une surprime pour les représentants les plus efficaces.

Il vous demande de compléter ce tableau à l'aide des informations suivantes :

- Le salaire de base de chaque représentant est de 1500 € brut
- La prime sur chiffre d'affaires est égale à 3 % du CA bois brut plus 3,5 % du CA Parquets plus 4 % du CA Accessoires
- Une surprime est versée selon les modalités suivantes : Si la prime est inférieure à 700 € la surprime est de zéro, si la prime est comprise entre 700 et 1 000 € la surprime est de 10 % et si la prime est supérieur à 1 000 € la surprime est de 20 %.

**Terminez ce tableau de telle sorte qu'il fasse apparaître : le salaire de base ; le montant de la prime, le montant de la surprime ; le montant de la prime total ; le salaire brut total du mois.**

### ➤ Tableau de l'indemnité kilométrique

L'indemnité kilométrique est égale à 26 centimes d'euros pour les voitures de plus de 10 CV fiscaux et de 20 centimes d'euros pour les voitures 10 CV fiscaux et moins.

**Terminez ce tableau en calculant le montant de l'indemnité kilométrique à verser à chaque représentant**

**Imprimer les deux tableaux ainsi que les formules de calcul des deux tableaux**

<b>Apprentissage programmé</b>	
<b>Société Jurarbois</b>	
<b>Savoir-faire</b>	• Programmer une fonction recherche
<b>Pré-requis</b>	• Avoir réalisé l'exercice programmé du dossier 5
<b>Supports</b>	• Dossier <b>Jurarbois</b> sur la clé USB u dans l'espace élève du disque dur ou du serveur
<b>Durée</b>	• 50 minutes

### Réalisation :

Dans ce dossier, nous allons saisir sur une feuille des informations concernant des salariés puis saisir sur une autre feuille un bulletin de salaire simplifié. Puis nous programmerons le bulletin de salaire pour qu'il se complète automatiquement à partir des données de la feuille salariés

-> Charger Excel et saisir les données suivantes :

	A	B	C	D	E	F	G	H
	Matricule	Nom	Qual.	Postes	Salaire	Prime Ancienneté	Kilomètre janvier	indemnité km
1								
2	101	BREDIN Jean	Cadre	PDG	3 200,00 €	150,00 €	1 820	345,80 €
3	102	BREDIN Julie	Cadre	Resp. administratif et financier	2 860,00 €	150,00 €	0	0,00 €
4	103	RATTE Jacques	Cadre	Resp. commercial	2 630,00 €	150,00 €	2 356	447,64 €
5	104	GAGEOT Joseph	Technicien	Resp. Production	2 220,00 €	70,00 €	698	132,62 €
6	105	JAQUIN Paul	Technicien	Technicien	1 610,00 €	70,00 €	587	111,53 €
7	106	DERUEL Luc	Technicien	Technicien	1 520,00 €	70,00 €	692	131,48 €
8	107	FILIBERT Marc	Technicien	Technicien	1 480,00 €	70,00 €	542	102,98 €
9	108	REMY Marie	Technicien	Technicien	1 420,00 €	70,00 €	608	115,52 €
10	109	LUCAS Jean	Technicien	Chauffeur livreur	1 400,00 €	70,00 €	0	0,00 €

### Remarque :

- La prime est de 150 € pour les cadres et 70 € pour les non cadres.
- L'indemnité kilométrique est de 0,19 € par km.

### ▪ Nommer la feuille : Salariés

-> Double cliquer l'onglet de la feuille **Feuil1** et saisir : **Salariés**



### ▪ Saisir le bulletin de salaire

-> Cliquer au bas de l'écran l'onglet **Feuil2**  
-> Saisir et mettre en forme le bulletin suivant

	A	B	C	D
1	<b>BULLETIN DE SALAIRE</b>			
2	Matricule : _____		Mois : _____	
3	Nom : _____			
4	Qualification : _____			
5	Poste : _____			
6				
7		Quantités	Taux horaire	Total
8	Salaire de base	151,67		
9	Prime d'ancienneté			
10				Brut
11		Taux	Base	Cotisations
12	Maladie, maternité	0,75		
13	Vieillesse	6,65		
14	CSG	7,50		
15	CRDS	0,50		
16	ASSEDIC	2,44		
17	Complémentaire	3,00		
18		Cotisations salariales		
19		Salaire net		
20		Indemnités		
21		Net à payer		

### ▪ Nommer la feuille : Salariés

-> Double cliquer l'onglet **Feuil2**  
-> Saisir : **Bulletin de salaire**



Nous disposons, à présent, de deux feuilles de calcul dans le même classeur :

- La feuille **Salariés** contient les informations concernant chaque salarié (**feuille source**),
- La feuille **Bulletin de salaire** doit recevoir des informations de la feuille salariés (**feuille destination**).

Nous souhaitons qu'en saisissant le matricule d'un salarié dans le bulletin de salaire (feuille destination) le nom, la qualification, le poste, le salaire de base, la prime d'ancienneté et l'indemnité kilométriques (qui se trouvent dans la feuille source) viennent automatiquement s'afficher dans le Bulletin de salaire.

Nous allons saisir, dans la feuille destination, la fonction **RECHERCHEV** dont la syntaxe est la suivante :

**Recherchev(donnée\_source;zone\_de\_recherche;n°\_colonne à copier)**

La recherche est réalisée à partir d'une donnée source (ici le matricule) qui est saisie dans la feuille destination. La fonction recherche le matricule dans la feuille Salariés (cellules A2 à G12) puis recopie la donnée qui se trouve dans la colonne dont le n° a été saisi dans la fonction.

Nous allons utiliser la fonction **RECHERCHEV** car les matricules sont classés verticalement dans la feuille Salariés. S'ils avaient été classés horizontalement nous aurions utilisé la fonction **RECHERCHEH**.

# 1. Programmer la fonction Recherche

-> Cliquer au bas de l'écran l'onglet de la feuille : **Bulletin de salaire**

## 1.1 Affichage automatique du nom

Pour contrôler le bon fonctionnement de notre formule, nous allons saisir un matricule dans le bulletin de salaire.

### ▪ Saisir la clé de recherche

-> Pointer la cellule : **B2** et saisir le matricule : **103**

### ▪ Saisir la formule


-> Pointer la cellule **B3** qui doit recevoir le nom du salarié

-> Cliquer l'onglet : **Formules**

-> Cliquer l'outil 

-> Cliquer la fonction **RECHERCHEV**

=> La fenêtre de paramétrage apparaît :



-> Déplacer la fenêtre pour voir le bulletin de salaire

-> Cliquer dans la zone : **Valeur\_cherchée**

-> Cliquer la cellule **B2** dans le Bulletin de salaire

-> Cliquer dans la zone : **Table\_matrice**

-> Cliquer l'onglet **Salariés** au bas de l'écran

-> Sélectionner par cliqué-glissé les cellules **A2 à H10**

-> Cliquer dans la zone : **No\_index\_col**

-> Saisir la valeur **2** (pour copier le contenu de la 2<sup>e</sup> colonne de la zone sélectionnée)

=> La fenêtre se présente ainsi :

-> **OK** ou **[Entrée]**

=> Le résultat apparaît dans la cellule et la formule dans la barre des formules :

BULLETIN DE SALAIRE			
Matricule :	103	Mois :	
Nom :	RATTE Jacques		
Qualification :			
Poste :			
	Quantités	Taux horaire	Total
Salaires de base	151,67		
Prime d'ancienneté			
	Brut		
	Taux	Base	Cotisations
Maladie, maternité	0,75		
Vieillesse	6,65		
CSG	7,50		
CRDS	0,50		
ASSEDIC	2,44		
Complémentaire	3,00		
	Cotisations salariales		
	Salaires net		
	Indemnités		
	Net à payer		

Nous allons saisir la formule de recherche de la qualification. Nous allons la saisir manuellement.

## 1.2 Affichage automatique de la qualification

-> Pointer la cellule **B4** qui doit recevoir la qualification

-> Taper : **=RECHERCHEV(**

-> Cliquer la cellule **B2** qui contient la donnée source

-> Taper le signe ;

-> Cliquer l'onglet **Salariés** et sélectionner les cellules **A2 à H10**

-> Taper le signe ;

-> Saisir la valeur **3** pour copier le contenu de la 3<sup>e</sup> colonne

-> Fermer la parenthèse )

-> **[Entrée]**

=> Le résultat apparaît dans la cellule :

BULLETIN DE SALAIRE			
Matricule :	103	Mois :	
Nom :	RATTE Jacques		
Qualification :	Cadre		
Poste :			
	Quantités	Taux horaire	Total
Salaires de base	151,67		
Prime d'ancienneté			
	Brut		
	Taux	Base	Cotisations
Maladie, maternité	0,75		
Vieillesse	6,65		
CSG	7,50		
CRDS	0,50		
ASSEDIC	2,44		
Complémentaire	3,00		
	Cotisations salariales		
	Salaires net		
	Indemnités		
	Net à payer		

Nous allons continuer la paramétrage des fonctions «Recherche» pour le poste, le salaire de base, la prime d'ancienneté et l'indemnité kilométrique au bas du bulletin

Nous avons étudié deux méthodes (Assistant et manuelle). Utilisez la méthode que vous préférez.

### 1.3 Affichage du poste

- > Pointer la cellule **B5** qui doit recevoir le poste
- > Programmer la fonction : **=RECHERCHEV(B2;Salariés!A2:H10;4)**

### 1.4 Affichage du salaire de base

- > Pointer la cellule **D8** qui doit recevoir le salaire de base
- > Programmer la fonction : **=RECHERCHEV(B2;Salariés!A2:H10;5)**

### 1.5 Affichage de la prime d'ancienneté

- > Pointer la cellule **D9** qui doit recevoir la prime
- > Programmer la fonction : **=RECHERCHEV(B2;Salariés!A2:H10;6)**

### 1.6 Affichage de l'indemnité de transport

- > Pointer la cellule **D20** qui doit recevoir l'indemnité
- > Programmer la fonction : **=RECHERCHEV(B2;Salariés!A2:H10;8)**

	A	B	C	D
1	<b>BULLETIN DE SALAIRE</b>			
2	<i>Matricule :</i>	103	<i>Mois :</i>	
3	<i>Nom :</i>	RATTE Jacques		
4	<i>Qualification :</i>	Cadre		
5	<i>Poste :</i>	Resp. commercial		
6				
7		<b>Quantités</b>	<b>Taux horaire</b>	<b>Total</b>
8	Salaire de base	151,67		2 630,00 €
9	Prime d'ancienneté			150,00 €
10			<b>Brut</b>	
11		<b>Taux</b>	<b>Base</b>	<b>Cotisations</b>
12	Maladie, maternité	0,75		
13	Vieillesse	6,65		
14	CSG	7,50		
15	CRDS	0,50		
16	ASSEDIC	2,44		
17	Complémentaire	3,00		
18		<b>Cotisations salariales</b>		
19			<b>Salaire net</b>	
20			<b>Indémnités</b>	447,64 €
21			<b>Net à payer</b>	



Il est possible d'attribuer un nom à la table dans laquelle se fait la recherche et de remplacer dans toutes les formules les références de la zone par son nom.

Nous avons programmé toutes les recherches automatiques. Nous allons à présent programmer les formules de calcul des taux horaires et du total brut. Les formules sont les suivantes :

#### Taux horaire :

- > **C8 =D8 /B8**

#### Total brut

- > **D10 =Somme(D8:D9)**

#### Les bases

- > **C12 =\$D\$10**

- > Recopier la formule sur les cellules **C13 à C17**

#### Cotisations

- > **D12 =C12\*B12/100**

- > Recopier la formule sur les cellules **D13 à D17**

#### Total des cotisations

- > **D18 =SOMME(D12 :D17)**

- > [Entrée]

#### Salaire net

- > **D19 =D10-D18**

#### Net à payer

- > **D21 =D19+D20**

=> Le bulletin terminé se présente ainsi :

	A	B	C	D
1	<b>BULLETIN DE SALAIRE</b>			
2	<i>Matricule :</i>	103	<i>Mois :</i>	
3	<i>Nom :</i>	=RECHERCHEV(B2;Salariés!A2:H10;2)		
4	<i>Qualification :</i>	=RECHERCHEV(B2;Salariés!A2:H10;3)		
5	<i>Poste :</i>	=RECHERCHEV(B2;Salariés!A2:H10;4)		
6				
7		<b>Quantités</b>	<b>Taux horaire</b>	<b>Total</b>
8	Salaire de base	151,67	=D8/B8	=RECHERCHEV(B2;Salariés!A2:H10;5)
9	Prime d'ancienneté			=RECHERCHEV(B2;Salariés!A2:H10;6)
10			<b>Brut</b>	=SOMME(D8:D9)
11		<b>Taux</b>	<b>Base</b>	<b>Cotisations</b>
12	Maladie, maternité	0,75	=\$D\$10	=B12*C12/100
13	Vieillesse	6,65	=\$D\$10	=B13*C13/100
14	CSG	7,5	=\$D\$10	=B14*C14/100
15	CRDS	0,5	=\$D\$10	=B15*C15/100
16	ASSEDIC	2,44	=\$D\$10	=B16*C16/100
17	Complémentaire	3	=\$D\$10	=B17*C17/100
18		<b>Cotisations salariales</b>		=SOMME(D12:D17)
19			<b>Salaire net</b>	=D10-D18
20			<b>Indémnités</b>	=RECHERCHEV(B2;Salariés!A2:H10;8)
21			<b>Net à payer</b>	=D19+D20

	A	B	C	D
1	<b>BULLETIN DE SALAIRE</b>			
2	<i>Matricule :</i>	103	<i>Mois :</i>	
3	<i>Nom :</i>	RATTE Jacques		
4	<i>Qualification :</i>	Cadre		
5	<i>Poste :</i>	Resp. commercial		
6				
7		<b>Quantités</b>	<b>Taux horaire</b>	<b>Total</b>
8	Salaire de base	151,67	17,34 €	2 630,00 €
9	Prime d'ancienneté			150,00 €
10			<b>Brut</b>	2 780,00 €
11		<b>Taux</b>	<b>Base</b>	<b>Cotisations</b>
12	Maladie, maternité	0,75	2 780,00 €	20,85 €
13	Vieillesse	6,65	2 780,00 €	184,87 €
14	CSG	7,50	2 780,00 €	208,50 €
15	CRDS	0,50	2 780,00 €	13,90 €
16	ASSEDIC	2,44	2 780,00 €	67,83 €
17	Complémentaire	3,00	2 780,00 €	83,40 €
18		<b>Cotisations salariales</b>		579,35 €
19			<b>Salaire net</b>	2 200,65 €
20			<b>Indémnités</b>	447,64 €
21			<b>Net à payer</b>	2 648,29 €

#### Attribuer éventuellement un format monétaire aux données

- > Cliquer l'outil

#### Imprimer le Bulletin de salaire

- > Cliquer l'outil

▪ **Afficher et imprimer les formules de calcul**

-> Cliquer l'onglet **Formules**

-> Cliquer l'outil  **Afficher les formules**

-> Réduire éventuellement la largeur des colonnes

-> Cliquer l'onglet **Mise en page**

-> Cliquer l'outil **Orientation**  et sélectionner **Paysage**

-> Cliquer l'outil 

=> Voir les formules affichées sur la page précédente

	A	B	C	D
1	<b>BULLETIN DE SALAIRE</b>			
2	<i>Matricule</i> : 103		<i>Mois</i> :	
3	<i>Nom</i> : =RECHERCHEV(B2;Salariés!A2:H10;2)			
4	<i>Qualification</i> : =RECHERCHEV(B2;Salariés!A2:H10;3)			
5	<i>Poste</i> : =RECHERCHEV(B2;Salariés!A2:H10;4)			
6				
7		<b>Quantités</b>	<b>Taux horaire</b>	<b>Total</b>
8	Salaire de base	151,67	=D8/B8	=RECHERCHEV(B2;Salariés!A2:H10;5)
9	Prime d'ancienneté			=RECHERCHEV(B2;Salariés!A2:H10;6)
10			<b>Brut</b>	=SOMME(D8:D9)
11		<b>Taux</b>	<b>Base</b>	<b>Cotisations</b>
12	Maladie, maternité	0,75	=\$D\$10	=B12*C12/100
13	Vieillesse	6,65	=\$D\$10	=B13*C13/100
14	CSG	7,5	=\$D\$10	=B14*C14/100
15	CRDS	0,5	=\$D\$10	=B15*C15/100
16	ASSEDIC	2,44	=\$D\$10	=B16*C16/100
17	Complémentaire	3	=\$D\$10	=B17*C17/100
18			<b>Cotisations salariales</b>	=SOMME(D12:D17)
19			<b>Salaire net</b>	=D10-D18
20			<b>Indemnités</b>	=RECHERCHEV(B2;Salariés!A2:H10;8)
21			<b>Net à payer</b>	=D19+D20

▪ **Afficher les résultats**

-> Cliquer l'outil  **Afficher les formules** pour afficher de nouveau les données

▪ **Sauvegarder le classeur**

-> Cliquer l'outil 

-> Sélectionner l'unité puis le dossier Jurarbois

-> Sauvegarder le classeur sous le nom : **Salaire**

-> Cliquer **Enregistrer**

▪ **Quitter le classeur**

-> Cliquer la case  fermeture du Classeur

<b>EXERCICE EN AUTONOMIE</b>	
<b>Société Strap SA</b>	
<b>Savoir-faire</b>	<ul style="list-style-type: none"> <li>• Programmer des fonctions conditionnelles</li> <li>• Programmer des fonctions recherches</li> </ul>
<b>Pré-requis</b>	<ul style="list-style-type: none"> <li>• Avoir effectué l'exercice programmé du dossier 6 et l'exercice en autonomie du dossier 2</li> </ul>
<b>Supports</b>	<ul style="list-style-type: none"> <li>• dossier <b>Strap Sa</b> sur la clé USB ou dans l'espace élève du disque dur ou du serveur</li> </ul>
<b>Durée</b>	<ul style="list-style-type: none"> <li>• 45 minutes</li> </ul>

**Travail à faire :**

1 - Charger le classeur **Facturation** réalisé dans le **Dossier 2**.

2 – Activer la feuille **Facture** et programmer les formules conditionnelles suivantes :

**Taux de remise**

Les modalités sont les suivantes :

taux = 0 % si brut < à 1000 €

taux = 5 % si brut >= à 1000 € et < à 2000 €

taux = 10 % si total brut >= à 2000

**Taux d'escompte**

Les modalités sont les suivantes :

- taux = 2 % si cellule escompte = Oui

- taux = 0 % si cellule escompte = Non

3 – Saisir sur une nouvelle feuille les articles suivants et nommer la feuille **Fichier articles**

4 - Programmer les fonctions **RECHERCHEV** dans la facture :

- Programmer l'affichage automatique de la **Désignation** dans la facture (réaliser la recherche à l'aide de la référence article)
- Programmer l'affichage automatique du **PUHT** dans la facture (réaliser la recherche à l'aide de la référence article)
- Intégrer les fonctions dans une fonction SI pour supprimer les messages d'erreurs,

	A	B	C
1	Référence	Désignation	PUHT
2	BETETA	Béton étanche (M3)	210,00 €
3	BETRES	Béton Résistant gel degel (M3)	220,00 €
4	BETSTA	Béton standard (M3)	190,00 €
5	CAMPOM	Camion pompe béton (Heure)	100,00 €
6	ETA50	Etais 50 (Mètre)	10,00 €
7	GRA16	Gravier rond 16 mn (tonne)	15,00 €
8	GRA4	Gravier rond 4 mn (tonne)	20,00 €
9	GRA8	Gravier rond 8 mn (tonne)	18,00 €
10	HEUBUL	Heure bulldozer	150,00 €
11	HEUCAMI	Heure camion	125,00 €
12	HEUCAMU	Heure Camuc	170,00 €
13	HEUSAL	Heure salarié	32,00 €
14	PIL30	Piliers d'angle 30*30	150,00 €
15	PIL50	Piliers d'angle 50*50	200,00 €
16	PLA27	Planche 27 (mètre)	8,00 €
17	SAB1	Sable 1 mn (tonne)	30,00 €
18	SAB2	Sable 2 mn (tonne)	28,00 €
19	SAB3	Sable 4 mn (tonne)	26,00 €
20	SABBLA	Sable blanc (tonne)	33,00 €
21	TREMET	Treilli métal (Mètre)	2,25 €

Solutions : cellule B12 =SI(A12="";"";RECHERCHEV(A12;'Fichier Articles'!\$A\$2:\$C\$21;2))

Explication : Si A12 est vide "" ; ne rien afficher "" dans B12 ; sinon faire la recherche et copier le contenu de la 2<sup>e</sup> colonne

cellule D12 =SI(A12="";"";RECHERCHEV(A12;'Fichier Articles'!\$A\$2:\$C\$21;3))

cellule E12 =SI(A12="";"";D12\*C12)

5 - Imprimer les formules de calcul et imprimer la facture vierge

6 - Sauvegarder le classeur

7 - Réaliser une facture :

71 - Copier la feuille Facture dans le classeur

72 - Nommer la feuille copiée : **Facture 257**

73 - Saisir les données suivantes dans la facture :

**Facture 257 du 15-10-2008 ; Règlement comptant, TVA à 19,6 %, pas d'escompte.**

Réf.	Désignation	Qté
BETSTA	Béton standard (M3)	4
GRA16	Gravier rond 16 mn (tonne)	4
HEUSAL	Heure salarié	20
HEUCAMI	Heure camion	4

74 - imprimer la feuille



Corrigé : Le net à payer doit être de **2 226,95 €**

8 - Sauvegarder le classeur et quitter Excel

<b>Dossier</b>  <b>6</b>	<b>Gestion des fiches de stocks</b>	
	Fonction Recherche	

<b>Mise en situation</b>	
<b>Société Morisse Bois et dérivés</b>	
<b>Savoir-faire</b>	<ul style="list-style-type: none"> <li>Paramétrer la gestion des fiches de stocks</li> </ul>
<b>Pré-requis</b>	<ul style="list-style-type: none"> <li>Avoir effectué les exercices du dossier 6</li> </ul>
<b>Supports</b>	<ul style="list-style-type: none"> <li>Dossier <b>Morisse bois</b> sur la Clé USB ou dans l'espace élève du disque dur ou du serveur</li> </ul>
<b>Durée</b>	<ul style="list-style-type: none"> <li>55 minutes</li> </ul>

### **Travail à faire :**

Vous travaillez pour la société **Morisse bois et dérivés**. Cette société est spécialisée dans le traitement et la commercialisation de bois sous différentes formes : **Bois brut** (planche, poutre, liteau etc.) ; **Parquets** (tout type de bois) ; **Accessoires** (vis, clous, colles etc.).

Elle souhaite informatiser la gestion de ses fiches de stocks. Le Directeur commercial, **Albert Gauchet**, vous remet le tableau récapitulatif des fournisseurs de la société.

N° frs	Raison sociale	Civilité	Responsable	Rue	CP	Ville	Téléphone
401001	FRAGRANT SA	Monsieur	DUPONT Pierre	3 rue de la Vieille Eglise	74000	ANNECY	04 50 78 58 68
401002	COURBON SARL	Madame	COURBOIN Lucie	118 rue de Genève	38000	GRENOBLE	04 76 48 45 42
401003	DE NEUVILLE	Monsieur	DE NEUVILLE Albert	4 rue A Briand	69000	LYON	04 78 32 35 39
401004	SOMANTRAN	Madame	GARROT Camille	19 av. de Loverchy	38000	GRENOBLE	04 72 45 86 62
401005	BUGET TRANSPORT	Monsieur	PERNOT Raymond	9 rue de la République	26000	VALENCE	04 75 42 12 12
401006	TERRANO GARAGE	Monsieur	ASSANS Jean	18 rue du Mont Blanc	74000	ANNECY	04 50 79 93 31
401007	APACHE CONSEIL	Monsieur	RIBOIRE Jean	5 rue des Capucins	26000	VALENCE	04 75 22 53 53
401008	LE DAUPHINE	Madame	JANNOT Louise	5 rue de la Gard	73000	CHAMBERY	04 42 72 72 73
401009	BOILEAU SA	Monsieur	BOILEAU Roger	18 rue de la Poste	69000	LYON	04 78 97 94 91
401010	TRINBERT SCI	Madame	TRINBERT Lucie	34 place Tabarly	69000	LYON	04 78 41 52 63

Il vous demande de concevoir une fiche de stock par produit. Elle devra faire apparaître les informations suivantes :

- **En-tête de la fiche**
  - Référence et désignation de l'article
  - Informations fournisseur : numéro fournisseur, Raison sociale, adresse, téléphone.

**Important :** L'entête de la fiche sera automatiquement complété lors de la saisie du numéro fournisseur dans la fiche de stock à partir des données contenues dans le fichier fournisseurs saisie précédemment.
- **Corps de la fiche**
  - La fiche doit permettre de gérer les stocks en quantité et en valeur,
  - Elle doit enregistrer tous les mouvements qui concernent l'article (entrées et sorties en valeur unitaire et en valeur totale),
  - Le stock est valorisé au coût moyen pondéré des stocks,
  - Les sorties sont évaluées au coût moyen unitaire pondéré.

Programmer toutes les formules de calcul

### **Imprimer**

- 1. le fichier fournisseurs,**
- 2. la fiche de stock vierge,**
- 3. les formules de calcul de la fiche de stock**

Tester votre fiche de stock avec les informations suivantes :

- Stock initial au 01/02/2008 : 110 articles valorisés à 15,24 €
- 01/02/2008 entrée en stock de 50 articles au prix unitaire de 16,77 €
- 02/02/2008 entrée en stock de 60 articles à au prix unitaire de 16,77 €
- 03/02/2008 Sortie du stock de 100 articles
- 04/02/2008 entrée en stock de 70 articles à au prix unitaire de 17,07 €
- 05/02/2008 Sortie du stock de 100 articles

### **Imprimer la fiche de stock**

<b>Dossier 7</b>	<b>7.1 - Tableaux d'amortissement linéaire d'une immobilisation</b>	
	Fonctions AMORLIN et AMORLINC	

<b>Apprentissage programmé Société Jurarbois</b>	
<b>Savoir-faire</b>	<ul style="list-style-type: none"> <li>Programmer les formules financières d'un tableau d'amortissement linéaire d'une immobilisations</li> </ul>
<b>Pré-requis</b>	<ul style="list-style-type: none"> <li>Savoir programmer une fonction mathématique (Dossier 5)</li> <li>Avoir étudié l'amortissement linéaire d'une immobilisation</li> </ul>
<b>Supports</b>	<ul style="list-style-type: none"> <li>Dossier <b>Jurarbois</b> sur la clé USB ou dans l'espace élève du disque dur ou du serveur</li> </ul>
<b>Durée</b>	<ul style="list-style-type: none"> <li>30 minutes</li> </ul>

**Réalisation :**

Dans ce dossier, nous allons concevoir et programmer le tableau d'amortissement linéaire d'une immobilisation :

	A	B	C	D
1	<b>Tableau d'amortissement linéaire immobilisation</b>			
2	Ref.	3895RF69	Désignation	Citroën C6
3		Coût d'achat	48 100,00 €	
4		Date achat	23/02/2008	
5		Date fin exercice comptable	31/12/2008	
6		VNC fin amortissement	0	
7		Durée	5	
8		Taux	20%	
9		Base de calcul	4	
10		Annuité	9 620,00 €	
11	Périodes	VNC Début période	Amortissement	VNC fin de période
12	1	48 100,00 €	8 203,72 €	39 896,28 €
13	2	39 896,28 €	9 620,00 €	30 276,28 €
14	3	30 276,28 €	9 620,00 €	20 656,28 €
15	4	20 656,28 €	9 620,00 €	11 036,28 €
16	5	11 036,28 €	9 620,00 €	1 416,28 €
17	6	1 416,28 €	1 416,28 €	0,00 €

Pour construire ce tableau d'amortissement nous allons utiliser les fonctions : =AMORLIN() et =AMORLINC().



**Amortissement linéaire d'une immobilisation**

**- AMORLIN(Coût\_achat;VNC\_fin\_période;durée)**

Calcule l'annuité d'un amortissement linéaire d'une immobilisation.

**VNC\_fin\_période** : représente la valeur résiduelle du bien au terme de l'amortissement.

**Durée** : représente le nombre de périodes pendant lesquelles le bien est amorti

**- AMORLINC(Coût\_achat;Date\_achat;Date\_fin\_exercice;VNC\_fin\_période;Durée\_amt;Taux\_amt;Base)**

Calcule l'amortissement du début de l'exercice à la date d'acquisition. En conséquence l'annuité calculée par AMORLINC doit être retiré de l'annuité calculée par AMORLIN.

**Taux\_amt** : représente le taux d'amortissement.

**Base** : représente la base annuelle à utiliser. Codes disponibles :

0	Base annuelle
1	360 jours (méthode NASD).
3	Réelle
3	365 jours par an
4	360 jours par an (méthode européenne)

▪ **Charger Excel puis saisir et mettre en forme le tableau suivant dans un nouveau classeur**

Les éléments nécessaires à la programmation des deux fonctions sont saisis dans le haut du tableau.

	A	B	C	D
1	<b>Tableau d'amortissement linéaire immobilisation</b>			
2	Ref.	3895RF69	Désignation	Citroën C6
3		Coût d'achat	48 100,00 €	
4		Date achat	23/02/2008	
5		Date fin exercice comptable	31/12/2008	
6		VNC fin amortissement	0	
7		Durée	5	
8		Taux		
9		Base de calcul	4	
10		Annuité		
11	Périodes	VNC Début période	Amortissement	VNC fin de période
12				
13				
14				
15				
16				
17				


# 1. Programmer les formules

## 1.1 Programmer le calcul du taux

- > Cliquer l'onglet : **Formules**
- > Cliquer la cellule **C8** et saisir la formule : **=1/C7**
- > Cliquer l'outil **%**

## 1.2 Programmer le calcul de l'annuité

- > Cliquer la cellule **C10**
- > Cliquer l'outil **Financier** puis la fonction **AMORLIN**
- > Programmer la formule de la façon suivante :
- > Cliquer **OK**
- => La formule est : **=AMORLIN(C3;C6;C7)**
- L'annuité est de **9 620,00 €**



## 1.3 Programmer les années

L'immobilisation sera amortie sur 5 ans, la 1<sup>re</sup> année étant prorata temporis nous devons prévoir 6 années.

- > Saisir les années 2008 à 2013 dans les cellules **A12 à A17**

## 1.4 Programmer la VNC de 2008

La VNC de 2008 est égale à la valeur d'achat saisie dans la cellule C3.


- > Cliquer la cellule **B12** et saisir la formule : **=C3**

Tableau d'amortissement linéaire immobilisation			
2	Réf. 3895RF69	Désignation	Citroën C6
3		Coût d'achat	48 100,00 €
4		Date achat	23/02/2008
5		Date fin exercice comptable	31/12/2008
6		VNC fin amortissement	0
7		Durée	5
8		Taux	20%
9		Base de calcul	4
10		Annuité	9 620,00 €
11	Périodes	VNC Début période	Amortissement fin de période
12	2008	48 100,00 €	
13	2009		
14	2010		
15	2011		
16	2012		
17	2013		

## 1.5 Programmer la 1<sup>re</sup> annuité

La 1<sup>re</sup> annuité est égale à l'annuité entière calculé dans la cellule C10, moins l'annuité prorata temporis calculé à l'aide de la fonction AMORLINC.

- > Cliquer la cellule **C12** et saisir la formule : **=C10-**
- > Cliquer l'outil **Financier** puis la fonction **AMORLINC**
- > Programmer la formule de la façon suivante :
- > **OK**
- => La formule est : **=C10-AMORLINC(C3;C4;C5;C6;C7;C8;C9)**
- L'amortissement est de **8 203,72 €**



## 1.6 Programmer la VNC de fin de période

La VNC de fin d'exercice est égale à la VNC de début d'exercice moins l'annuité de l'année.

- > Cliquer la cellule **D12**
- > Saisir la formule : **=B12-C12**

## 1.7 Programmer la VNC de la 2<sup>e</sup> ligne

La VNC de la 2<sup>e</sup> ligne est égale à la VNC de fin de période de la ligne précédente.

- > Cliquer la cellule **B13** et saisir la formule : **=D12**

## 1.8 Programmer la 2<sup>e</sup> annuité

La 2<sup>e</sup> annuité est égale à l'annuité entière calculée dans la cellule C10. Nous allons la programmer en référence absolue pour la recopier vers le bas.

- > Cliquer la cellule **C13** et saisir la formule : **=\$C\$10**

Tableau d'amortissement linéaire immobilisation			
2	Réf. 3895RF69	Désignation	Citroën C6
3		Coût d'achat	48 100,00 €
4		Date achat	23/02/2008
5		Date fin exercice comptable	31/12/2008
6		VNC fin amortissement	0
7		Durée	5
8		Taux	20%
9		Base de calcul	4
10		Annuité	9 620,00 €
11	Périodes	VNC Début période	Amortissement fin de période
12	2008	48 100,00 €	8 203,72 € 39 896,28 €
13	2009	39 896,28 €	9 620,00 € 30 276,28 €
14	2010		
15	2011		
16	2012		
17	2013		

## 1.9 Programmer la VNC de fin de période

La VNC de fin d'exercice est égale à la VNC de début d'exercice moins l'annuité de l'année.

- > Cliquer la cellule **D13** et saisir la formule : **=B13-C13**

Les formules programmées pour la 2<sup>e</sup> ligne sont les mêmes à programmer pour les 3 lignes suivantes. Nous pouvons donc les recopier vers le bas sur les 3 lignes suivantes ou les saisir manuellement.

### 1.10 Recopier les formules

- > Sélectionner les cellules B13 à D13
- > Cliquer glisser le bouton de recopie de la ligne sur la cellule D16
- => Le tableau se présente ainsi :

	A	B	C	D
1	<b>Tableau d'amortissement linéaire immobilisation</b>			
2	Réf. 3895RF69		Désignation	Citroën C6
3		Coût d'achat	48 100,00 €	
4		Date achat	23/02/2008	
5		Date fin exercice comptable	31/12/2008	
6		VNC fin amortissement	0	
7		Durée	5	
8		Taux	20%	
9		Base de calcul	4	
10		Annuité	9 620,00 €	
11	Périodes	VNC Début période	Amortissement	VNC fin de période
12	2008	48 100,00 €	8 203,72 €	39 896,28 €
13	2009	39 896,28 €	9 620,00 €	30 276,28 €
14	2010	30 276,28 €	9 620,00 €	20 656,28 €
15	2011	20 656,28 €	9 620,00 €	11 036,28 €
16	2012	11 036,28 €	9 620,00 €	1 416,28 €
17	2013			

### 1.11 Programmer la VNC de la dernière ligne

La VNC de la dernière ligne est égale à la VNC de fin de période de la ligne précédente.

- > Cliquer la cellule B17
- > Saisir la formule : =D16

### 1.12 Programmer la dernière annuité

La dernière annuité est égale à la VNC initiale de la dernière ligne.

- > Cliquer la cellule C17
- > Saisir la formule : =B17

### 1.13 Programmer la VNC de fin de période

La VNC de fin d'exercice est égale à la VNC de début d'exercice moins l'annuité de l'année.

- > Cliquer la cellule D17
- > Saisir la formule : =B17-C17
- => Le tableau terminé se présente ainsi :

	A	B	C	D
1	<b>Tableau d'amortissement linéaire immobilisation</b>			
2	Réf. 3895RF69		Désignation	Citroën C6
3		Coût d'achat	48 100,00 €	
4		Date achat	23/02/2008	
5		Date fin exercice comptable	31/12/2008	
6		VNC fin amortissement	0	
7		Durée	5	
8		Taux	20%	
9		Base de calcul	4	
10		Annuité	9 620,00 €	
11	Périodes	VNC Début période	Amortissement	VNC fin de période
12	2008	48 100,00 €	8 203,72 €	39 896,28 €
13	2009	39 896,28 €	9 620,00 €	30 276,28 €
14	2010	30 276,28 €	9 620,00 €	20 656,28 €
15	2011	20 656,28 €	9 620,00 €	11 036,28 €
16	2012	11 036,28 €	9 620,00 €	1 416,28 €
17	2013	1 416,28 €	1 416,28 €	0,00 €

## 2 Imprimer le tableau d'amortissement

- > Cliquer l'outil 

## 3 Imprimer les formules de calcul

### ▪ Afficher et imprimer les formules de calcul

- > Cliquer l'onglet **Formules**
- > Cliquer l'outil  **Afficher les formules**
- > Réduire éventuellement la largeur des colonnes
- > Cliquer l'onglet **Mise en page**
- > Cliquer l'outil **Orientation** 
- > Sélectionner **Paysage**
- > Cliquer l'outil 

	A	B	C	D
1	<b>Tableau d'amortissement linéaire immobilisation</b>			
2	Réf. 3895RF69		Désignation	Citroën C6
3		Coût d'achat	48100	
4		Date achat	39501	
5		Date fin exercice comptable	39813	
6		VNC fin amortissement	0	
7		Durée	5	
8		Taux	=1/C7	
9		Base de calcul	4	
10		Annuité	=AMORLIN(C3;C6;C7)	
11	Périodes	VNC Début période	Amortissement	VNC fin de période
12	2008	=C3	=C10-AMORLINC(C3;C4;C5;C6;C7;C8;C9)	=B12-C12
13	2009	=D12	=SC\$10	=B13-C13
14	2010	=D13	=SC\$10	=B14-C14
15	2011	=D14	=SC\$10	=B15-C15
16	2012	=D15	=SC\$10	=B16-C16
17	2013	=D16	=B17	=B17-C17

### ▪ Afficher les résultats

- > Cliquer l'outil  **Afficher les formules** pour afficher de nouveau les données

### ▪ Sauvegarder le classeur

- > Cliquer l'outil 
- > Sélectionner l'unité puis le dossier **Jurabois**
- > Sauvegarder le classeur sous le nom : **Tableau amortissement linéaire**
- > Cliquer **Enregistrer**

### ▪ Quitter le classeur

- > Cliquer la case  fermeture du Classeur

<b>Dossier 7</b>	<b>7.2 - Tableaux d'amortissement dégressif d'une immobilisation</b>	
	Fonction VDB	

<b>Apprentissage programmé Société Jurarbois</b>	
<b>Savoir-faire</b>	<ul style="list-style-type: none"> <li>• Programmer des formules financières dans le cadre d'un tableau d'amortissement dégressif d'une immobilisation</li> </ul>
<b>Pré-requis</b>	<ul style="list-style-type: none"> <li>• Avoir réalisé les exercices du dossier 5</li> <li>• Savoir programmer une fonction mathématique</li> <li>• Avoir étudié l'amortissement dégressif d'une immobilisation</li> </ul>
<b>Supports</b>	<ul style="list-style-type: none"> <li>• Dossier <b>Jurarbois</b> sur la clé USB ou dans l'espace élève du disque dur ou du serveur</li> </ul>
<b>Durée</b>	<ul style="list-style-type: none"> <li>• 30 minutes</li> </ul>

**Réalisation :**

Dans ce dossier, nous allons concevoir et programmer le tableau d'amortissement dégressif d'une immobilisation :

Les éléments nécessaires à la programmation des fonctions sont saisis en haut du tableau.

Pour construire le tableau d'amortissement dégressif nous allons utiliser la fonction : **=VDB( )**.

	A	B	C	D
1	<b>Tableau d'amortissement dégressif immobilisation</b>			
2	Ref. 3895RF69		Désignation	Citroën C6
3				
4		Coût achat :	48 100,00 €	
5		Valeur résiduelle	0	
6		Durée :	5	
7		Taux :	20%	
8	Périodes	VNC Début période	Amortissement	VNC fin de période
9	1	48 100,00 €	19 240,00 €	28 860,00 €
10	2	28 860,00 €	11 544,00 €	17 316,00 €
11	3	17 316,00 €	6 926,40 €	10 389,60 €
12	4	10 389,60 €	5 194,80 €	5 194,80 €
13	5	5 194,80 €	5 194,80 €	0,00 €

**Amortissement dégressif d'une immobilisation**

**VDB(coût\_achat;valeur\_rés;durée;période\_début;période\_fin;facteur;valeur\_log)**  
 Calcule l'amortissement d'un bien en utilisant la méthode de l'amortissement dégressif avec changement de taux en fin de période.

**valeur\_rés** = valeur résiduelle en fin d'amortissement,  
**durée** = durée de vie utile du bien,  
**période\_début** = n° de la 1<sup>re</sup> période  
**période\_fin** = n° de la période suivante  
**facteur** = taux de l'amortissement dégressif. La valeur par défaut est 2 (méthode de l'amortissement dégressif à taux double). Modifier sa valeur si vous ne souhaitez pas utiliser la méthode de l'amortissement dégressif à taux double.  
**valeur\_log** = Vrai ou Faux, indique s'il faut utiliser l'amortissement linéaire lorsqu'il est > à celui obtenu par l'amortissement dégressif. (VRAI => n'applique pas l'amortissement linéaire, FAUX => amortissement linéaire)

-> Charger Excel puis saisir et mettre en forme le tableau suivant dans un nouveau classeur

**1 Programmer les formules**

**1.1 Programmer le calcul du taux**

- > Cliquer la cellule C7
- > Programmer la formule suivante : **=1/C6**
- > Cliquer l'outil **%**

**1.2 Programmer les années**

- L'immobilisation sera amortie sur 5 ans, nous devons prévoir 5 années.
- > Cliquer la cellule A9
  - > Saisir périodes 1 à 5

	A	B	C	D
1	<b>Tableau d'amortissement dégressif immobilisation</b>			
2	Ref. 3895RF69		Désignation	Citroën C6
3				
4		Coût achat	48 100,00 €	
5		Valeur résiduelle	0	
6		Durée	5	
7		Taux	20%	
8	Périodes	VNC Début période	Amortissement	VNC fin de période
9	1			
10	2			
11	3			
12	4			
13	5			

### 1.3 Programmer la VNC de la période 1


La VNC de période 1 est égale à la valeur d'achat saisie dans la cellule C4.

-> Cliquer la cellule B9 et saisir la formule : =C4

### 1.4 Programmer la 1<sup>re</sup> annuité

La 1<sup>re</sup> annuité est calculée à l'aide de la fonction =VDB().

-> Cliquer la cellule C9


-> Cliquer l'outil  puis la fonction VDB

-> Programmer la formule de la façon suivante :

-> OK

=> La formule est : =VDB(C4;C5;C6;A9-1;A9)

L'amortissement est de 19 240,00 €



	A	B	C	D
1	<b>Tableau d'amortissement dégressif immobilisation</b>			
2	Ref. 3895RF69	Désignation Citroën C6		
3				
4		Coût achat	48 100,00 €	
5		Valeur résiduelle	0	
6		Durée	5	
7		Taux	20%	
8	Périodes	VNC Début période	Amortissement	VNC fin de période
9	1	48 100,00 €	19 240,00 €	28 860,00 €
10	2			
11	3			
12	4			
13	5			

### 1.5 Programmer la VNC de fin de période

La VNC de fin d'exercice est égale à la VNC de début d'exercice moins l'annuité de l'année.

-> Cliquer la cellule D9 et saisir la formule : =B9-C9

### 1.6 Programmer la VNC de la 2<sup>e</sup> ligne

La VNC de la 2<sup>e</sup> ligne est égale à la VNC de fin de période de la ligne précédente.

-> Cliquer la cellule B10 et saisir la formule : =D9

### 1.7 Programmer la 2<sup>e</sup> annuité

La formule est identique à la formule du dessus. Nous ne pouvons pas la recopier car certaines cellules ne sont pas en référence absolue. Programmer la formule manuellement en plaçant des \$ au endroit indiqués.

-> Cliquer la cellule C10 et saisir la formule :

=VDB(\$C\$4;\$C\$5;\$C\$6;A10-1;A10)

	A	B	C	D	E
1	<b>Tableau d'amortissement dégressif immobilisation</b>				
2	Ref. 3895RF69	Désignation Citroën C6			
3					
4		Coût achat	48 100,00 €		
5		Valeur résiduelle	0		
6		Durée	5		
7		Taux	20%		
8	Périodes	VNC Début période	Amortissement	VNC fin de période	
9	1	48 100,00 €	19 240,00 €	28 860,00 €	
10	2	28 860,00 €	11 544,00 €	17 316,00 €	
11	3				
12	4				
13	5				

### 1.8 Programmer la VNC de fin de période

La VNC de fin d'exercice est égale à la VNC de début d'exercice moins l'annuité de l'année.

-> Cliquer la cellule D10 et saisir la formule : =B10-C10

Les formules programmées pour la 2<sup>e</sup> ligne sont les mêmes à programmer pour les 3 lignes suivantes. Nous pouvons donc les recopier vers le bas sur les 3 lignes suivantes ou les saisir manuellement.

### 1.9 Recopier les formules

-> Sélectionner les cellules B10 à D10

-> Cliquer glisser le bouton de recopie de la ligne sur la cellule D13

=> Le tableau terminé se présente ainsi :

	A	B	C	D
1	<b>Tableau d'amortissement dégressif immobilisation</b>			
2	Ref. 3895RF69	Désignation Citroën C6		
3				
4		Coût achat	48 100,00 €	
5		Valeur résiduelle	0	
6		Durée	5	
7		Taux	20%	
8	Périodes	VNC Début période	Amortissement	VNC fin de période
9	1	48 100,00 €	19 240,00 €	28 860,00 €
10	2	28 860,00 €	11 544,00 €	17 316,00 €
11	3	17 316,00 €	6 926,40 €	10 389,60 €
12	4	10 389,60 €	5 194,80 €	5 194,80 €
13	5	5 194,80 €	5 194,80 €	0,00 €

## 2 Imprimer le tableau d'amortissement

-> Cliquer l'outil 

## 3 Imprimer les formules de calcul

### ▪ Afficher et imprimer les formules de calcul

-> Cliquer l'onglet **Formules**

-> Cliquer l'outil  **Afficher les formules**

-> Réduire éventuellement la largeur des colonnes

-> Cliquer l'onglet **Mise en page**

-> Cliquer l'outil 

	A	B	C	D
1	<b>Tableau d'amortissement dégressif immobilisation</b>			
2	Ref. 3895RF69		Désignation	Citroën C6
3				
4		Coût achat	48100	
5		Valeur résiduelle	0	
6		Durée	5	
7		Taux	=1/C6	
8	<b>Périodes</b>	<b>VNC Début période</b>	<b>Amortissement</b>	<b>VNC fin de période</b>
9	1	=C4	=VDB(C4;C5;C6;A9-1;A9;2;FAUX)	=B9-C9
10	2	=D9	=VDB(\$C\$4;\$C\$5;\$C\$6;A10-1;A10;2;FAUX)	=B10-C10
11	3	=D10	=VDB(\$C\$4;\$C\$5;\$C\$6;A11-1;A11;2;FAUX)	=B11-C11
12	4	=D11	=VDB(\$C\$4;\$C\$5;\$C\$6;A12-1;A12;2;FAUX)	=B12-C12
13	5	=D12	=VDB(\$C\$4;\$C\$5;\$C\$6;A13-1;A13;2;FAUX)	=B13-C13

### ▪ Afficher les résultats

-> Cliquer l'outil  **Afficher les formules** pour afficher de nouveau les données

### ▪ Sauvegarder le classeur

-> Cliquer l'outil 

-> Sélectionner l'unité puis le dossier **Jurarbois**

-> Sauvegarder le classeur sous le nom : **Tableau amortissement dégressif**

-> Cliquer **Enregistrer**

### ▪ Quitter le classeur

-> Cliquer la case  fermeture du Classeur

<b>Dossier 7</b>	<b>7.3 - Tableaux d'amortissement d'un emprunt</b>	
	Fonctions : VPM, INTPER, PRINCPER	

<b>Apprentissage programmé Société Jurarbois</b>	
<b>Savoir-faire</b>	<ul style="list-style-type: none"> <li>Programmer des formules financières dans le cadre d'un tableau d'amortissement d'emprunts</li> </ul>
<b>Pré-requis</b>	<ul style="list-style-type: none"> <li>Savoir programmer une fonction mathématique (Dossier 5)</li> <li>Avoir étudié l'amortissement d'un emprunt</li> </ul>
<b>Supports</b>	<ul style="list-style-type: none"> <li>Dossier <b>Jurarbois</b> sur la clé USB ou dans l'espace élève du disque dur ou du serveur</li> </ul>
<b>Durée</b>	<ul style="list-style-type: none"> <li>30 minutes</li> </ul>

**Réalisation :**

Au cours de cette étape, nous allons concevoir et programmer le tableau d'amortissement d'un emprunt (extrait) :

Pour construire le tableau d'amortissement linéaire nous allons utiliser les fonctions :

**=VPM( ) ; =INTPER( ) ; =PRINCPER( )**

	A	B	C	D	E	F
1	<b>Tableau d'emprunt</b>					
2	Capital :	120 000,00 €				
3	Taux	7,00%				
4	Durée	48				
5	Mensualité	2 873,55 €				
6						
7	Périodes	Capital Début période	Intérêt	Amortissement	Mensualité	Capital fin de période
8	1	120 000,00	700,00	2 173,55	2 873,55	117 826,45
9	2	117 826,45	687,32	2 186,23	2 873,55	115 640,22
10	3	115 640,22	674,57	2 198,98	2 873,55	113 441,24
11	4	113 441,24	661,74	2 211,81	2 873,55	111 229,43



**Amortissement d'un emprunt à annuités ou mensualités constantes**

Les formules **VPM()**, **INTPER()** et **PRINCPER()** permettent de programmer un tableau d'emprunt.

- **VPM(taux;nbre\_périodes;Capital)** => calcule la mensualité ou l'annuité.
- **INTPER(taux;n° période;nbre périodes;Capital)** => calcule l'intérêt des mensualités ou annuités.
- **PRINCPER(taux;n° période;nbre périodes;Capital)** => calcule l'amortissement des mensualités ou annuités.

**Annuité ou mensualité ?**

- **Annuité** : saisir un taux d'intérêt annuel (8 %) et saisir un nombre d'annuités (4 pour 4 ans)
- **Mensualité** : saisir un taux d'intérêt mensuel (Taux annuel/12) puis saisir un nombre de mensualités (48 pour 4 ans) ou saisir un taux d'intérêt annuel puis saisir un nombre de mensualité (48 pour 4 ans). Mais dans ce dernier cas le taux devra être divisé par 12 dans la fonction. (Voir exemple illustré ci-dessous)

-> Charger Excel puis saisir et mettre en forme le tableau suivant dans un nouveau classeur

Le prêt sera amorti sur 48 mois, prévoir 48 mensualités.

-> Cliquer la cellule **A8**

-> Saisir dans les cellules **A8 à A56** les périodes de **1 à 48**

	A	B	C	D	E	F
1	<b>Tableau d'emprunt</b>					
2	Capital :	120 000,00 €				
3	Taux	7,00%				
4	Durée	48				
5	Mensualité					
6						
7	Périodes	Capital Début période	Intérêt	Amortissement	Mensualité	Capital fin de période
8	1					
9	2					
10	3					
11	4					
12	5					
13	6					
14	7					
15	8					
16	9					
17	10					

**1 Programmer les formules**

**1.1 Programmer le calcul d'une mensualité**

La mensualité est calculée par la fonction **=VPM()**

-> Cliquer la cellule **B5**

-> Saisir la formule : **=-VPM(B3/12;B4;B2)**

=> L'amortissement est de **2 873,55 €** :

	A	B	C	D	E	F
1	<b>Tableau d'emprunt</b>					
2	Capital :	120 000,00 €				
3	Taux	7,00%				
4	Durée	48				
5	Mensualité	2 873,55 €				
6						
7	Périodes	Capital Début période	Intérêt	Amortissement	Mensualité	Capital fin de période
8	1	120 000,00				
9	2					
10	3					
11	4					
12	5					

**1.2 Programmer le capital de début de période**

Il est égal au montant saisi dans la cellule **B2**.

-> Cliquer la cellule **B8** et saisir la formule : **=B2**

### 1.3 Programmer le calcul de l'intérêt

L'intérêt est calculé par la fonction =INTPER()

-> Cliquer la cellule C8 et saisir la formule : =INTPER(\$B\$3/12;A8;\$B\$4;\$B\$2)

### 1.4 Programmer le capital de l'amortissement

L'intérêt est calculé par la fonction =PRINCPER()

-> Cliquer la cellule D8 et saisir la formule :  
=-PRINCPER(\$B\$3/12;A8;\$B\$4;\$B\$2)

### 1.5 Calcul de la mensualité

La mensualité est égale à l'amortissement plus l'intérêt.

-> Cliquer la cellule E8 et saisir la formule : =C8+D8

### 1.6 Calcul du capital en fin de période

Il est égal au capital début de période moins l'amortissement de la période.

-> Cliquer la cellule F8 et saisir la formule : =B8-D8

### 1.7 Calcul du capital en début de période

Il est égal au capital de fin de période de la ligne précédente.

-> Cliquer la cellule B9 et saisir la formule : =F8

Les formules de calcul des cellules C9 à F9 sont les mêmes que celles de la ligne 8. Nous allons donc les recopier sur la ligne 9.

### 1.8 Recopier le contenu de la 1<sup>re</sup> ligne sur la seconde ligne

-> Sélectionner les cellules C8 à F8 puis cliquer-glisser le bouton de copie de la zone sur la cellule F9  
=> Le tableau se présente ainsi et le contenu de la ligne 9 peut être recopié à présent sur les 46 lignes du dessous.

### 1.9 Recopier la 2<sup>e</sup> ligne sur les lignes du bas

-> Sélectionner les cellules A9 à F9 puis cliquer-glisser le bouton de copie de la zone sur la cellule F55  
=> Le tableau terminé se présente ainsi : (La ligne 55 se termine par un capital de fin de période de 0)

Tableau d'emprunt						
1						
2	Capital :	120 000,00 €				
3	Taux	7,00%				
4	Durée	48				
5	Mensualité	2 873,55 €				
6						
7	Périodes	Capital Début période	Intérêt	Amortissement	Mensualité	Capital fin de période
8	1	120 000,00	700,00	2 173,55	2 873,55	117 826,45
9	2	117 826,45				
10	3					
11	4					
12	5					

Tableau d'emprunt						
1						
2	Capital :	120 000,00 €				
3	Taux	7,00%				
4	Durée	48				
5	Mensualité	2 873,55 €				
6						
7	Périodes	Capital Début période	Intérêt	Amortissement	Mensualité	Capital fin de période
8	1	120 000,00	700,00	2 173,55	2 873,55	117 826,45
9	2	117 826,45	687,32	2 186,23	2 873,55	115 640,22
10	3	115 640,22	674,57	2 198,98	2 873,55	113 441,24
11	4	113 441,24	661,74	2 211,81	2 873,55	111 229,43
12	5	111 229,43	648,84	2 224,71	2 873,55	109 004,72
51	44	14 119,69	82,36	2 791,18	2 873,55	11 328,51
52	45	11 328,51	66,08	2 807,47	2 873,55	8 521,04
53	46	8 521,04	49,71	2 823,84	2 873,55	5 697,20
54	47	5 697,20	33,23	2 840,32	2 873,55	2 856,88
55	48	2 856,88	16,67	2 856,88	2 873,55	0,00

## 2 Imprimer le tableau d'amortissement

-> Cliquer l'outil 

## 3 Imprimer les formules de calcul

### ▪ Afficher et imprimer les formules de calcul

-> Cliquer l'onglet Formules puis cliquer l'outil  Afficher les formules

-> Cliquer l'outil 

### ▪ Afficher les résultats

-> Cliquer l'outil  Afficher les formules pour afficher de nouveau les données

### ▪ Sauvegarder le classeur

-> Cliquer l'outil 

-> Sélectionner l'unité puis le dossier Jurarbois puis saisir le nom : Tableau d'emprunt

-> Cliquer Enregistrer

### ▪ Quitter le classeur

-> Cliquer la case  fermeture du Classeur

<b>EXERCICE EN AUTONOMIE Société Strap SA</b>	
<b>Savoir-faire</b>	• Concevoir un tableau d'amortissement d'emprunt
<b>Pré-requis</b>	• Avoir effectué les exercices 7.1 à 7.3
<b>Supports</b>	• Dossier <b>Strap SA</b> sur la clé USB ou dans l'espace élève du disque dur ou du serveur
<b>Durée</b>	• 35 minutes

**Travail à faire :**

La société a faite l'acquisition d'une voiture de livraison le 1<sup>er</sup> février 2008. L'achat concerne le modèle **Trafic de Renault** dont les caractéristiques sont les suivantes :

- Prix d'achat TTC : 35 880,00 €
- TVA : 5 880,00 €
- Prix d'achat HT : 30 000,00 €
- Financement par un prêt 24 000,00 €
- Durée 36 mois

**A – Tableau d'emprunt**

- 1 - Saisir et mettre en forme le tableau d'amortissement du prêt :
- 2 - Programmer les fonctions qui permettent d'obtenir les mensualités :  
**B8 =-VPM(B6/12;B7;B5)**  
**C11 =-INTPER(\$B\$6/12;A11;\$B\$7;\$B\$5)**  
**D11 =-PRINCPER(\$B\$6/12;A11;\$B\$7;\$B\$5)**
- 3 - Imprimer le tableau (La colonne de fin de période vous servira pour contrôler l'exactitude de vos données)
- 4 - imprimer les formules de calcul des 3 premières lignes
- 5 – Nommer la feuille **Tableau d'emprunt**
- 6 - Sauvegarder le classeur dans le dossier Strap SA sous le nom : **Renault Trafic**

	A	B	C	D	E	F
1	<b>Tableau d'emprunt</b>					
2	Banque : CIC Lyonnaise banque		Date emprunt :	01/02/2008		
3	Immo. : Trafic Renault		1re échéance :	01/05/2008		
4			Durée :	36 mensualités		
5	Capital	24 000,00 €				
6	Taux	8,00%				
7	Durée	36				
8	Mensualité					
9						
10	Périodes	Capital Début période	Intérêt	Amortissement	Mensualité	Capital fin de période
11	1					23 407,93
12	2					22 811,91
13	3					22 211,91
14	4					21 607,92
15	5					20 999,90
16	6					20 387,83
17	7					19 771,67
18	8					19 151,41
19	9					18 527,01
20	10					17 898,46
21	11					17 265,71
22	12					16 628,74
23	13					15 987,52
24	14					15 342,03
25	15					14 692,24
26	16					14 038,12
27	17					13 379,63
28	18					12 716,76
29	19					12 049,46
30	20					11 377,72
31	21					10 701,50
32	22					10 020,77
33	23					9 335,50
34	24					8 645,66
35	25					7 951,23
36	26					7 252,16
37	27					6 548,44
38	28					5 840,02
39	29					5 126,88
40	30					4 408,99
41	31					3 686,31
42	32					2 958,81
43	33					2 226,47
44	34					1 489,24
45	35					747,09
46	36					0,00

**B – Tableau d'amortissement dégressif**

- 1 - Saisir et mettre en forme le tableau d'amortissement dégressif qui correspond au véhicule Trafic acheté ci-dessus sur une nouvelle feuille du classeur :
- 2 - Programmer les fonctions qui permettent d'obtenir le tableau d'amortissement dégressif :  
**C10 =VDB(\$B\$7;\$D\$5;\$D\$6;A10-1;A10)**
- 3 - Imprimer le tableau (La colonne VNC de fin de période vous servira pour contrôler l'exactitude de vos données)
- 4 - Imprimer les formules de calcul des deux premières lignes
- 5 – Nommer la feuille **Tableau amortissement**
- 5 - Sauvegarder le classeur dans le dossier Jurarbois sous le nom : **Renault trafic**
- 6 - Fermer le classeur

	A	B	C	D
1	<b>Tableau d'amortissement dégressif immobilisation</b>			
2				
3	Réf.	Trafic 02-2008	Désignation	Renault Trafic
4				
5	Prix achat TTC	35 880,00 €	Valeur résiduelle	0
6	TVA	5 880,00 €	Durée	10
7	Prix d'achat HT	30 000,00 €	Taux	10%
8				
9	Périodes	VNC Début période	Amortissement	VNC fin de période
10	1			24 000,00 €
11	2			19 200,00 €
12	3			15 360,00 €
13	4			12 288,00 €
14	5			9 830,40 €
15	6			7 864,32 €
16	7			5 898,24 €
17	8			3 932,16 €
18	9			1 966,08 €
19	10			0,00 €

<b>Dossier</b> <b>7</b>	<b>Tableau d'amortissement de prêt</b> <b>Tableau d'amortissement linéaire</b>	
	Fonctions : VPM, INTPER, PRINCPER, AMORLIN et AMORLINC	

<b>Mise en situation</b> <b>Société Morisse Bois et dérivés</b>	
<b>Savoir-faire</b>	• Concevoir un tableau d'amortissement d'emprunt
<b>Pré-requis</b>	• Avoir effectué les exercices 7.1 à 7.3
<b>Supports</b>	• Dossier <b>Morisse bois</b> sur la clé USB ou dans l'espace élève du disque dur ou du serveur
<b>Durée</b>	• 50 minutes

### **Travail à faire :**

La société Morisse bois et dérivés a réalisé l'acquisition d'une nouvelle **machine à bois à commande numérique HEREDIA** dont la référence est **MA-NUM-2008** et dont le prix TTC est de 118 000 €. (TVA à 19,6 %.) La date d'acquisition est le 01/05/2008.

L'exercice comptable de la société va du 01/01/n au 31/12/n

#### **▪ Tableau d'amortissement**

Il vous demande de réaliser le tableau d'amortissement linéaire de la machine sachant qu'elle sera amortie sur une durée de 6 ans.

Nommer la feuille : *Tableau amortissement linéaire*

Imprimer le tableau d'amortissement de la machine et les formules de calcul.

#### **▪ Etude de l'emprunt**

Le responsable financier envisage de financer cette acquisition par un apport de 20 662.20 € et le reste serait financé par un prêt bancaire.

Monsieur Gauchet a téléphoné aux trois banques de la société qui lui ont communiqué leurs meilleurs propositions.

<b>Banque CIC</b>	<b>Banque BNP Paribas</b>	<b>Banque Crédit Agricole</b>
<i>Prêt sur 5 ans au taux de 7,3 %</i>	<i>Prêt sur 6 ans au taux de 5.9 %</i>	<i>Prêt sur 7 ans au taux de 5,1 %</i>

Il vous demande de lui indiquer pour chaque prêt les informations suivantes :

- Montant des mensualités
- Total des intérêts à payer sur la période

Attention : chaque tableau d'emprunt doit être réalisé sur une feuille différente dont les noms seront différents également

Il vous demande :

- d'imprimer les trois tableaux de prêt
- de lui indiquer quel est le prêt le plus intéressant et de justifier votre choix par les avantages et inconvénients

Sauvegarder votre travail dans le dossier **Morisse** sous le nom sera **Etude investissement amortissement prêt**